

## E.7 STDATM

This subroutine computes the 1976 standard atmosphere. It is used in program FRICTION. It covers an altitude range from sea level to 86 kilometers (282,152 ft.). The results are found in either English or metric units depending on the value of one of the input flags. The 1976 and 1962 standard atmospheres are identical for the first 51 kilometers above sea level.

*Method of Computation*

Given the geometric altitude  $Z_{in}$  (in dimensions of either meters or feet), convert to kilometers. The geopotential altitude  $H$  is then found from:

$$H = \frac{Z}{1 + \frac{Z}{r_0}}$$

where  $r_0 = 6356.766$  kilometers (the radius of the Earth in kilometers) and  $Z = C_1 Z_{in}$ , where  $C_1 = 0.001$  if  $Z_{in}$  is in meters, and  $C_1 = 0.0003048$  if  $Z_{in}$  is in feet. The 1962 standard atmosphere used a much more complicated and slightly more accurate relationship.

The inverse relation is given by

$$Z = \frac{H}{1 - \frac{H}{r_0}}$$

Once the geopotential altitude is found, the temperature is computed. The standard day temperature profile is defined by seven layers, where within each layer the temperature is found by the linear relation ( $T$  is given in degrees Kelvin):

$$T = T_{b_i} + L_{m_i}(H - H_{b_i})$$

and  $T_{b_i}$ ,  $L_{m_i}$  and  $H_{b_i}$  are the values at the base of the particular layer. The following table defines these constants, as well as the ratio of pressure to sea level pressure, which is also needed.

$i$	$H_{b_i}$ (Km)	$T_{b_i}$ (°K)	$L_{m_i}$ (°K/Km)	$P/Psl$	$Z$ (ft.)
1	0.	288.15	-6.5	1.0	0
2	11.	216.65	0.0	$2.2336 \times 10^{-1}$	36,152.
3	20.	216.65	+1.0	$5.4032 \times 10^{-2}$	65,824
4	32.	228.65	+2.8	$8.5666 \times 10^{-3}$	105,518
5	47.	270.65	0.0	$1.0945 \times 10^{-3}$	155,348
6	51.	270.65	-2.8	$6.6063 \times 10^{-4}$	168,676
7	71.	214.65	-2.0	$3.9046 \times 10^{-5}$	235,571
-	82.	-	-	-	282,152

Once the temperature is determined, the pressure is computed using the hydrostatics equation and the perfect gas law. The resulting formulas are:

$$\frac{P}{P_{sl}} = \frac{P_b}{P_{sl}} \left( \frac{T_b}{T} \right)^{\frac{K}{L_m}} \quad L_m \neq 0$$

$$\frac{P}{P_{sl}} = \frac{P_b}{P_{sl}} e^{-\frac{K(H-H_b)}{T_b}} \quad L_m = 0$$

where  $K = \frac{g_0 M_0}{R^*} = 34.163195$  in consistent units. The remaining fundamental property is the density, which is found using the equation of state as:

$$\frac{\rho}{\rho_{sl}} = \frac{P/P_{sl}}{T/T_{sl}}$$

Additional parameters of interest in aerodynamics are:

i) The speed of sound

$$a = a_{sl} \sqrt{\frac{T}{T_{sl}}}$$

ii) The coefficient of viscosity, found from Sutherland's Law:

$$\mu = \frac{\beta \cdot T^{3/2}}{T + S}$$

where  $S = 110.4^\circ\text{K}$  and  $\beta$  depends on the system of units and is defined below.

iii) The Reynolds number per unit length and Mach:

$$\frac{R_e}{M \cdot L} = \frac{\rho a}{\mu}$$

iv) The actual temperature, pressure and density:

$$T = T_{sl} \left( \frac{P}{P_{sl}} \right)$$

$$P = P_{sl} \left( \frac{P}{P_{sl}} \right)$$

$$\rho = \rho_{sl} \left( \frac{\rho}{\rho_{sl}} \right)$$

and  $\nu$ ) the dynamic pressure normalized by the Mach number:

$$\frac{q}{M^2} = \frac{\gamma}{2} P = .7P$$

The sea level properties and other required constants are defined in the following table.

	<u>Metric</u>	<u>English</u>
$T_{sl}$	288.15 °K	518.67° R
$P_{sl}$	1.01325x10 <sup>5</sup> N/m <sup>2</sup>	2116.22 lb/ft <sup>2</sup>
$\rho_{sl}$	1.2250 Kg/m <sup>3</sup>	0.0023769 slugs/ft <sup>3</sup>
$a_{sl}$	340.294 m/sec	1116.45 ft/sec
$\mu_{sl}$	1.7894x10 <sup>-5</sup> Kg/m/sec	0.37373x10 <sup>-6</sup> slugs/ft/sec
$\beta$	1.458x10 <sup>-6</sup> Kg/m/sec/K <sup>1/2</sup>	3.0450963x10 <sup>-8</sup> slugs/ft/sec/K <sup>1/2</sup>

The ratio of specific heats,  $\gamma$ , is defined to be 1.40.

User instructions: the comments in the subroutine define the input and output argument list. If the maximum altitude is exceeded, the program returns a non zero value of the validity flag.

```

subroutine stdatm(z,t,p,r,a,mu,ts,rr,pp,rm,qm,kd,kk)
c
c ***** 1976 STANDARD ATMOSPHERE SUBROUTINE *****
c
c Mason's BASIC program, converted to FORTRAN - Sept. 1, 1989
c
c kd - = 0 - metric units
c      <> 0 - English units
c
c kk - 0 - good return
c      1 - error: altitude out of table,
c          do not use output
c
c z - input altitude, in feet or meters (depending on kd)
c
c output:
c
c          units:      metric      English
c t - temp.           deg K       deg R
c p - pressure        N/m^2      lb/ft^2
c r - density         Kg/m^3      slugs/ft^3
c a - speed of sound  m/sec      ft/sec
c mu - viscosity      Kg/m/sec    slug/ft/sec
c
c ts - t/t at sea level
c rr - rho/rho at sea level
c pp - p/p at sea level
c
c rm - Reynolds number per      Re/M/m      Re/M/ft
c      Mach per unit of length
c qm - dynamic pressure/Mach^2  N/m^2      lb/ft^2

```

```
c
  real k, h, mu, ml
  KK = 0
  K = 34.163195
  C1 = 3.048E-04
  IF (KD .eq. 0) go to 1240
  TL = 518.67
  PL = 2116.22
  RL = .0023769
  AL = 1116.45
  ML = 3.7373E-07
  BT = 3.0450963E-08
  GO TO 1260

1240 TL = 288.15
  PL = 101325
  RL = 1.225
  C1 = .001
  AL = 340.294
  ML = 1.7894E-05
  BT = 1.458E-06

1260 H = C1 * Z / (1 + C1 * Z / 6356.766)
  IF (H .gt. 11.0) go to 1290
  T = 288.15 - 6.5 * H
  PP = (288.15 / T) ** ( - K / 6.5)
  GO TO 1420

1290 IF (H .gt. 20.0) go to 1310
  T = 216.65
  PP = .22336 * EXP ( - K * (H - 11) / 216.65)
  GO TO 1420

1310 IF (H .gt. 32.0) go to 1330
  T = 216.65 + (H - 20)
  PP = .054032 * (216.65 / T) ** K
  GO TO 1420

1330 IF (H .gt. 47.0) go to 1350
  T = 228.65 + 2.8 * (H - 32)
  PP = .0085666 * (228.65 / T) ** (K / 2.8)
  GO TO 1420

1350 IF( H .gt. 51.0) go to 1370
  T = 270.65
  PP = .0010945 * EXP ( - K * (H - 47) / 270.65)
  GO TO 1420

1370 IF (H .gt. 71.) go to 1390
  T = 270.65 - 2.8 * (H - 51)
  PP = .00066063 * (270.65 / T) ** ( - K / 2.8)
  GO TO 1420

1390 IF (H .gt. 84.852) THEN
      kk = 1
      write(6,200) H
      return
  END IF
```

## E-30 Applied Computational Aerodynamics

---

```
T = 214.65 - 2 * (H - 71)
PP = 3.9046E-05 * (214.65 / T) ** ( - K / 2)

1420 RR = PP / (T / 288.15)
MU = BT * T**1.5 / (T + 110.4)
TS = T / 288.15
A = AL * SQRT (TS)
T = TL * TS
R = RL * RR
P = PL * PP
RM = R * A / MU
QM = .7 * P

200 format(' Out of Table in StdAtm- too high !'//
1         4x,'H =',f12.3,' > 84.852 km'//

return
end
```

The following sample program and output can be used to validate your subroutine:

```
c main program to check stdatm

c loop is done twice to get output
c suitable to include in text(80 col)

c w.h. mason, Feb. 27, 1994

real mu

kd = 1

write(6,90)

do 10 i = 1,21
z = 5000.*(i-1)

call stdatm(z,t,p,r,a,mu,ts,rr,pp,rm,qm,kd,kk)

if (kk .ne. 0) then
write(6,120)
stop
endif

write(6,100) z,t,p,r,a,mu
10 continue

write(6,110)
do 20 i = 1,21
z = 5000.*(i-1)

call stdatm(z,t,p,r,a,mu,ts,rr,pp,rm,qm,kd,kk)

if (kk .ne. 0) then
write(6,160)
stop
endif
```

```

      write(6,120) z,ts,rr,pp,rm,qm
20 continue

90 format(/3x,'1976 Standard Atmosphere'//
1      3x,' alt      T      P      Rho',
2      2x,'      a      Mu',
4      /3x,' (ft)    (deg R) (psf)    (s/ft^3)',
5      2x,' (f/s)    (slugs/ft/sec)')
100 format(3x,f9.1,f8.2,f8.2,e12.4,f8.2,e12.4)
110 format(/3x,'1976 Standard Atmosphere'//
1      3x,' alt      T/Tsl  R/Rsl',
2      2x,' P/Psl  Re/M/ft  q/M^2',
4      /3x,' (ft)',34x,'(lb/ft^2)')
120 format(3x,f9.1,3f7.4,e10.3,f10.4)
160 format(/4x,'error in return code from stdatm - pgm stops')
      stop
      end

```

## Sample output:

## 1976 Standard Atmosphere

alt (ft)	T (deg R)	P (psf)	Rho (s/ft <sup>3</sup> )	a (f/s)	Mu (slugs/ft/sec)
0.0	518.67	2116.22	0.2377E-02	1116.45	0.3737E-06
5000.0	500.84	1760.88	0.2048E-02	1097.10	0.3637E-06
10000.0	483.03	1455.60	0.1756E-02	1077.40	0.3534E-06
15000.0	465.22	1194.79	0.1496E-02	1057.36	0.3430E-06
20000.0	447.42	973.28	0.1267E-02	1036.93	0.3324E-06
25000.0	429.62	786.34	0.1066E-02	1016.10	0.3217E-06
30000.0	411.84	629.67	0.8907E-03	994.85	0.3107E-06
35000.0	394.06	499.35	0.7382E-03	973.14	0.2995E-06
40000.0	389.97	393.13	0.5873E-03	968.08	0.2969E-06
45000.0	389.97	309.45	0.4623E-03	968.08	0.2969E-06
50000.0	389.97	243.61	0.3639E-03	968.08	0.2969E-06
55000.0	389.97	191.80	0.2865E-03	968.08	0.2969E-06
60000.0	389.97	151.03	0.2256E-03	968.08	0.2969E-06
65000.0	389.97	118.93	0.1777E-03	968.08	0.2969E-06
70000.0	392.25	93.73	0.1392E-03	970.90	0.2984E-06
75000.0	394.97	73.99	0.1091E-03	974.26	0.3001E-06
80000.0	397.69	58.51	0.8571E-04	977.62	0.3018E-06
85000.0	400.42	46.35	0.6743E-04	980.95	0.3035E-06
90000.0	403.14	36.78	0.5315E-04	984.28	0.3052E-06
95000.0	405.85	29.23	0.4196E-04	987.59	0.3070E-06
100000.0	408.57	23.27	0.3318E-04	990.90	0.3087E-06

## 1976 Standard Atmosphere

alt (ft)	T/Tsl	R/Rsl	P/Psl	Re/M/ft	q/M <sup>2</sup> (lb/ft <sup>2</sup> )
0.0	1.0000	1.0000	1.0000	0.710E+07	1481.3538
5000.0	0.9656	0.8617	0.8321	0.618E+07	1232.6129
10000.0	0.9313	0.7386	0.6878	0.535E+07	1018.9235
15000.0	0.8969	0.6295	0.5646	0.461E+07	836.3538
20000.0	0.8626	0.5332	0.4599	0.395E+07	681.2936
25000.0	0.8283	0.4486	0.3716	0.337E+07	550.4373
30000.0	0.7940	0.3747	0.2975	0.285E+07	440.7683

## E-32 Applied Computational Aerodynamics

---

35000.0	0.7598	0.3106	0.2360	0.240E+07	349.5441
40000.0	0.7519	0.2471	0.1858	0.191E+07	275.1887
45000.0	0.7519	0.1945	0.1462	0.151E+07	216.6139
50000.0	0.7519	0.1531	0.1151	0.119E+07	170.5264
55000.0	0.7519	0.1205	0.0906	0.934E+06	134.2600
60000.0	0.7519	0.0949	0.0714	0.736E+06	105.7186
65000.0	0.7519	0.0747	0.0562	0.579E+06	83.2541
70000.0	0.7563	0.0586	0.0443	0.453E+06	65.6079
75000.0	0.7615	0.0459	0.0350	0.354E+06	51.7925
80000.0	0.7668	0.0361	0.0276	0.278E+06	40.9574
85000.0	0.7720	0.0284	0.0219	0.218E+06	32.4446
90000.0	0.7772	0.0224	0.0174	0.171E+06	25.7445
95000.0	0.7825	0.0177	0.0138	0.135E+06	20.4621
100000.0	0.7877	0.0140	0.0110	0.107E+06	16.2903

STOP